



Unsung Names on Planes

With the help of text via [procedural textures](#) in Arma 3 2.12, we started to exploit the possibility to display pilot names below the canopy of an aircraft. This document documents the setup steps in case for the Unsung A-7 Corsair II.

Utility functions

After some experimentation with the text on textures technology, we decided to place some utility functions into the Unsung missilebox that should make it easier to integrate the names below the canopy on Unsung planes.

The first function `fn_setNameOnPlane.sqf` makes some calls on the look of the text, like font and size being used.

```
/**
 * Author:
 * - TeTeT
 *
 * Description:
 * - Write the name of a unit as text on a texture
 *
```

```

* Parameter(s):
*   _unit - the unit that gives the name
*   _vehicle - the vehicle where the text is displayed
*   _selection - the selection on the vehicle model on which the name will be written
*
* Returns:
*   _unit
**/
params ["_unit", "_vehicle", "_selection"];
private _name = format ["%1 %2", [_unit, "displayNameShort"] call BIS_fnc_rankParams,
    [name _unit] call uns_mbox_fnc_abbreviateName];
// Maybe make this configurable in the future from settings found in the vehicle?
_vehicle setObjectTextureGlobal [_selection,
    format ['#(rgb,512,512,3)text(1,1,"RobotoCondensed",0.2,"#00000000", "#111111",%1)', _name]];

_unit;

```

During testing it became apparent that long names don't look to well, so a second utility function was introduced: `fn_abbreviateName.sqf`:

```

/**
* Author:
* - TeTeT
*
* Description:
* - Abbreviate all names but the last one
*
* Parameter(s):
*   _name - a string with space separated names
*
* Returns:
*   _newName - the abbreviated name
**/
params [["_name", ""]];

private _newName = _name;
private _nameArray = _name splitString " ";

```

```

if (count _nameArray > 0) then {
    // systemChat str _nameArray;
    private _lastName = _nameArray deleteAt (count _nameArray - 1);
    _newName = "";
    {
        private _firstLetter = _x select [0, 1];
        _newName = format ["%1 %2", _newName, _firstLetter];
    } forEach _nameArray;
    if (count _newName > 0) then {
        _newName = format ["%1 %2", _newName select [1], _lastName];
    } else {
        _newName = _lastName;
    };
};
_newName;

```

Both functions are part of the Unsung Missilebox's function library and can be called via:

```

uns_mbox_fnc_setNameOnPlane
uns_mbox_fnc_abbreviateName

```

For use in the init eventhandler of a plane in Unsung, the `fn_common_init.sqf` function was adjusted. It is part of the init of any Unsung plane:

```

// Modified by TeTeT

params [["_vehicle", ObjNull]];


_vehicle setVariable ["Afterburner", 0, false];
_vehicle setVariable ["uns_fuel tanks", 1, false];
_vehicle setVariable ["ELS_Active", 0, false]; // arm FN

if (not isNull driver _vehicle) then {
    [driver _vehicle, _vehicle, "NameLeft"] call uns_mbox_fnc_setNameOnPlane;
};

```

Graphical walk-through


1. Setup a selection in the model in blender




Ensure that the Z-Bias flag is set on the plane over the regular plane body, so z-fighting is kept to a minimum.

Assign a transparent texture to the selection, in the case of A-7 we used the `glass_ca.paa` of the model.

2. Check the alignment of Text on the selection in blender



3. Export the A-7 from blender to p3d



4. Add the selection to the model section in Visual Studio Code

```
class CfgModels
{
    class Default
    { ...
    };
    class uns_A7_corsair: Default
    {
        skeletonName="uns_A7_corsair_Skeleton";
        sections[]=
        {
            "body",
            "wings",
            "gear",
            "misc",
            "L svetlo",
            "zbytek",
            "sklo predni l",
            "podsvit pristroju",
            "vrtule blur",
            "zasleh",
            "d_num1","d_num2",          //usaf decals
            "d_num3","d_num4","d_num5",    //navy decals
            "Hit_AVionics", "Hit_Glass1", "Hit_Glass2", "Hit_Hull",
            "Hit_LAileron", "Hit_LCElevator", "Hit_LCRudder", "Hit_RAileron",
            "NameLeft"
        };
    class Animations
```


5. Setup an eventhandler in config.cpp for the A-7

```
class Eventhandlers: EventHandlers
{
    class unsung
    {
        init = "_this spawn uns_a7_fnc_idnum_USAF; _this spawn uns_mbox_fnc_common_init;";
        fired = "_this spawn uns_mbox_fnc_fall";
        engine = "_this call bis_fnc_aircraftFoldingWings;";
        gear = "_this call bis_fnc_aircraftFoldingWings;";
        hit = "_this call bis_fnc_planeAiEject";
        GetIn = "[_this # 2, _this # 0, 'NameLeft'] call uns_mbox_fnc_setNameOnPlane;";
    };
};
```

6. Setup a hiddenSelection in config.cpp for the A-7

```
//model
displayName = "A-7D Corsair II";
model = "\uns_A7\uns_A7_corsair.p3d";
picture = "\uns_A7\ico\a7_ca.paa";
icon = "\uns_A7\ico\a7Map_ca.paa";
mapSize = 17;
hiddenSelections[] =
{
    "body",
    "wings",
    "gear",
    "misc",
    "d_num1", //USAF Tail number
    "d_num2", //USAF Tail letters
    "d_num3", //Navy nose and wing number
    "d_num4", //Navy nose and wing number
    "d_num5", //Navy nose and wing number
    "NameLeft"
};
hiddenSelectionsTextures[] =
```

7. Build pbo via Mikero PboProject



8. Test empty Cockpit in Arma 3



9. Test player in Cockpit in Arma 3



Appendix

Model.cfg excerpt:

```
class uns_A7_corsair: Default
{
    skeletonName="uns_A7_corsair_Skeleton";
    sections[]=
    {
        "body",
        "wings",
        "gear",
        "misc",
        "L svetlo",
        "zbytek",
        "sklo predni l",
        "podsvit pristroju",
        "vrtule blur",
        "zasleh",
        "d_num1","d_num2",          //usaf decals
    }
}
```



```

        "d_num3", "d_num4", "d_num5",          //navy decals
        "Hit_Avionics", "Hit_Glass1", "Hit_Glass2", "Hit_Hull",
        "Hit_LAileron", "Hit_LCElevator", "Hit_LCRudder", "Hit_RAileron",
"Hit_RElevator",
        "NameLeft"
};

```

Config.cpp excerpt:

```

//model
    displayName = "A-7D Corsair II";
    model = "\uns_A7\uns_A7_corsair.p3d";
    picture = "\uns_A7\ico\a7_ca.paa";
    icon = "\uns_A7\ico\a7Map_ca.paa";
    mapSize = 17;
    hiddenSelections[] =
    {
        "body",
        "wings",
        "gear",
        "misc",
        "d_num1",          //USAF Tail number
        "d_num2",          //USAF Tail letters
        "d_num3",          //Navy nose and wing number
        "d_num4",          //Navy nose and wing number
        "d_num5",          //Navy nose and wing number
        "NameLeft"
    };

```

```

class Eventhandlers: EventHandlers
{
    class unsung
    {
        init = "_this spawn uns_a7_fnc_idnum_USAF; _this spawn

```

```
uns_mbox_fnc_common_init;";
//      fired = "_this spawn uns_mbox_fnc_fall";
      engine = "_this call bis_fnc_aircraftFoldingWings;";
      gear = "_this call bis_fnc_aircraftFoldingWings;";
      hit = "_this call bis_fnc_planeAiEject";
      GetIn = "[_this # 2, _this # 0, 'NameLeft'] call
uns_mbox_fnc_setNameOnPlane;";
      };
};
```